UNIVERSIDAD POLITECNICA DE VALENCIA

DEPARTAMENTO DE INGENIERIA MECANICA Y DE MATERIALES

ELEMENTOS FINITOS

(E.T.S.I.I.V)

IMPLEMENTACION COMPUTACIONAL DEL METODO EF

LECCION 11.- IMPLEMENTACION DEL TRIANGULO ISO P

J. L. OLIVER
Dr. Ingeniero Industrial

Valencia, 2005

001 PECULIARIDADES PROPIAS DE TRIANGULOS CURSO 2004-5

- Special Gauss quadrature rules
- Computation of Jacobian and shape function x-y derivatives

001 CUADRATURA DE GAUSS PARA TRIANGULOS CURSO 2004-5
CONDICIONES ESTABILIDAD NUMERICA

SIMETRIA TRIANGULAR

> If the sample point $(\zeta_1, \zeta_2, \zeta_3)$ is present in a Gauss integration rule with weight w, then all other points obtainable by permuting the three triangular coordinates arbitrarily must appear in that rule, and have the same weight.

This rule guarantees that the result of the quadrature process will not depend on element node numbering.[1] If ζ_1, ζ_2, and ζ_3 are different, condition (24.1) forces six equal-weight sample points to be present in the rule, because $3! = 6$. If two triangular coordinates are equal, the six points coalesce to three, and the condition forces three equal-weight sample points to be present. Finally, if the three coordinates are equal (which can only happen for the centroid $\zeta_1 = \zeta_2 = \zeta_3 = 1/3$), the six points coalesce to one.[2]

It follows that the number of sample points in triangle Gauss quadrature rules must be of the form $6i + 3j + k$, where i and j are nonnegative integers and k is 0 or 1. Consequently there are no rules with 2, 5 or 8 points.

ACEPTABILIDAD NUMERICA

> All sample points must be inside the triangle (or on the triangle boundary) and all weights must be positive.

The reasons for the stability conditions (24.2) cannot be covered in an elementary course. They are automatically satisfied by all Gauss product rules for quadrilaterals, and so it was not necessary to call attention to them. On the other hand, for triangles there are Gauss rules with as few as 4 and 6 points that violate those conditions.

A rule is said to be of degree n if it integrates exactly all polynomials in the triangular coordinates of order n or less when the Jacobian determinant is constant, and there is at least one polynomial of order $n + 1$ that is not exactly integrated by the rule.



AREA DEL TRAINGULO ES CONSTANTE

$$A = \int_{\Omega^{(e)}} d\Omega^{(e)} = \tfrac{1}{2}\det \begin{bmatrix} 1 & 1 & 1 \\ x_1 & x_2 & x_3 \\ y_1 & y_2 & y_3 \end{bmatrix} = \tfrac{1}{2}\big[(x_2y_3 - x_3y_2) + (x_3y_1 - x_1y_3) + (x_1y_2 - x_2y_1)\big]$$

REGLA 1 PUNTO

$$\frac{1}{A}\int_{\Omega^{(e)}} F(\zeta_1, \zeta_2, \zeta_3)\, d\Omega^{(e)} \approx F(\tfrac{1}{3}, \tfrac{1}{3}, \tfrac{1}{3})$$

This rule is depicted in Figure 24.1(a). It has degree 1, meaning that it integrates exactly up to linear polynomials in triangular coordinates. For example, $F = 4 - \zeta_1 + 2\zeta_2 - \zeta_3$ is exactly integrated by (24.3).

REGLAS 3 PUNTOS

$$\frac{1}{A}\int_{\Omega^{(e)}} F(\zeta_1, \zeta_2, \zeta_3)\, d\Omega^{(e)} \approx \tfrac{1}{3}F(\tfrac{2}{3}, \tfrac{1}{6}, \tfrac{1}{6}) + \tfrac{1}{3}F(\tfrac{1}{6}, \tfrac{2}{3}, \tfrac{1}{6}) + \tfrac{1}{3}F(\tfrac{1}{6}, \tfrac{1}{6}, \tfrac{2}{3})$$

$$\frac{1}{A}\int_{\Omega^{(e)}} F(\zeta_1, \zeta_2, \zeta_3)\, d\Omega^{(e)} \approx \tfrac{1}{3}F(\tfrac{1}{2}, \tfrac{1}{2}, 0) + \tfrac{1}{3}F(0, \tfrac{1}{2}, \tfrac{1}{2}) + \tfrac{1}{3}F(\tfrac{1}{2}, 0, \tfrac{1}{2})$$

These are depicted in Figures 24.1(b) and (c), respectively. Both rules are of degree 2; that is, exact up to quadratic polynomials in the triangular coordinates. For example, the function $F = 6 + \zeta_1 + 3\zeta_3 + \zeta_2^2 - \zeta_3^2 + 3\zeta_1\zeta_3$ is integrated exactly by either rule. Formula (24.6) is called the *midpoint* rule.

REGLAS DE 6 Y 7 PUNTOS

Six and Seven Point Rules. There is a 4-point rule of degree 3 but it has a negative weight, which violates the stability condition (24.2). There are no symmetric rules with 5 points. The next useful rules have six and seven points. There is a 6-point rule of degree 4 and a 7-point rule of degree 5, which integrate exactly up to quartic and quintic polynomials, respectively. The 7-point rule includes the centroid as sample point. The abcissas and weights are expressable as rational combinations of square roots of integers and fractions. The expressions are listed in the *Mathematica* implementation shown in Figure 24.3. The rule configurations are depicted in Figures 24.1(d) and (e).



If the triangle has variable metric, as in the curved-side 6-node triangle geometries pictured in Figure 24.2, the foregoing formulas need adjustment because the element of area $d\Omega^{(e)}$ becomes a function of position. Consider the more general case of an isoparametric element with n nodes and shape functions N_i.

DIFERENCIAL DE AREA EL ELEMENTOS

$$d\Omega^{(e)} = J\, d\zeta_1\, d\zeta_2\, d\zeta_3, \qquad J = \tfrac{1}{2}\det \begin{bmatrix} 1 & 1 & 1 \\ \sum_{i=1}^{n} x_i \frac{\partial N_i}{\partial \zeta_1} & \sum_{i=1}^{n} x_i \frac{\partial N_i}{\partial \zeta_2} & \sum_{i=1}^{n} x_i \frac{\partial N_i}{\partial \zeta_3} \\ \sum_{i=1}^{n} y_i \frac{\partial N_i}{\partial \zeta_1} & \sum_{i=1}^{n} y_i \frac{\partial N_i}{\partial \zeta_2} & \sum_{i=1}^{n} y_i \frac{\partial N_i}{\partial \zeta_3} \end{bmatrix}$$

Here J is the Jacobian determinant, which plays the same role as J in the isoparametric quadrilaterals. If the metric is simply defined by the 3 corners, as in Figure 24.1, the geometry shape functions are $N_1 = \zeta_1$, $N_2 = \zeta_2$ and $N_3 = \zeta_3$. Then the foregoing determinant reduces to that of (24.4), and $J = A$ everywhere. But for general geometries $J = J(\zeta_1, \zeta_2, \zeta_3)$, and the triangle area A cannot be factored out of the integration rules.

REGLA 1 PUNTO

$$\int_{\Omega^{(e)}} F(\zeta_1, \zeta_2, \zeta_3)\, d\Omega^{(e)} \approx J(\tfrac{1}{3}, \tfrac{1}{3}, \tfrac{1}{3})\, F(\tfrac{1}{3}, \tfrac{1}{3}, \tfrac{1}{3})$$

REGLAS 3 PUNTOS

$$\int_{\Omega^{(e)}} F(\zeta_1, \zeta_2, \zeta_3)\, d\Omega^{(e)} \approx \tfrac{1}{3} J(\tfrac{1}{2}, \tfrac{1}{2}, 0)\, F(\tfrac{1}{2}, \tfrac{1}{2}, 0) + \tfrac{1}{3} J(0, \tfrac{1}{2}, \tfrac{1}{2})\, F(0, \tfrac{1}{2}, \tfrac{1}{2}) + \tfrac{1}{3} J(\tfrac{1}{2}, 0, \tfrac{1}{2})\, F(\tfrac{1}{2}, 0, \tfrac{1}{2})$$

```
TrigGaussRuleInfo[{rule_,numer_},point_]:= Module[
 {zeta,p=rule,i=point,g1,g2,info=Null},
  If [p== 1, info={{1/3,1/3,1/3},1}];
  If [p==-3, zeta={1/2,1/2,1/2}; zeta[[i]]=0; info={zeta,1/3}];
  If [p== 3, zeta={1/6,1/6,1/6}; zeta[[i]]=2/3; info={zeta,1/3}];
  If [p== 6,
     If [i<=3, g1=(8-Sqrt[10]+Sqrt[38-44*Sqrt[2/5]])/18;
         zeta={g1,g1,g1}; zeta[[i]]=1-2*g1;
         info={zeta,(620+Sqrt[213125-53320*Sqrt[10]])/3720}];
     If [i>3,  g2=(8-Sqrt[10]-Sqrt[38-44*Sqrt[2/5]])/18;
         zeta={g2,g2,g2}; zeta[[i-3]]=1-2*g2;
         info={zeta,(620-Sqrt[213125-53320*Sqrt[10]])/3720}]];
  If [p== 7,
     If [i==1,info={{1/3,1/3,1/3},9/40} ];
     If [i>1&&i<=4,zeta=Table[(6-Sqrt[15])/21,{3}];
         zeta[[i-1]]=(9+2*Sqrt[15])/21;
         info={zeta,(155-Sqrt[15])/1200}];
     If [i>4,  zeta=Table[(6+Sqrt[15])/21,{3}];
         zeta[[i-4]]=(9-2*Sqrt[15])/21;
         info={zeta,(155+Sqrt[15])/1200}]];
  If [numer, Return[N[info]], Return[Simplify[info]]];
 ];
```

The five rules of Figures 24.1–2 are implemented in a *Mathematica* module called `TrigGaussRuleInfo`, which is listed in Figure 24.3. It is called as

`{{zeta1,zeta2,zeta3},w} = TrigGaussRuleInfo[{rule,numer},point]` (24.10)

The module has 3 arguments: `rule`, `numer` and `i`. The first two are grouped in a two-item list. Argument `rule`, which can be 1, 3, −3, 6 or 7, defines the integration formula as follows: `Abs[rule]` is the number of sample points. Of the two 3-point choices, if `rule` is −3 the midpoint rule is picked, else if +3 the 3-interior point rule is chosen. Logical flag `numer` is set to `True` or `False` to request floating-point or exact information, respectively

Argument `point` is the index of the sample point, which may range from 1 through `Abs[rule]`.

The module returns the two-level list $\{\{\zeta_1, \zeta_2, \zeta_3\}, w\}$, where ζ_1, ζ_2, ζ_3 are the triangular coordinates of the sample point, and w is the integration weight. For example, the call `TrigGaussRuleInfo[{3,False},1]` returns `{{2/3,1/6,1/6},1/3}`. If `rule` is not 1, 3, −3, 6 or 7, the module returns `Null`.

001 CALCULO DE LAS DERIVADAS PARCIALES CURSO 2004-5

MODELO ELEMENTO CUADRATICO DE 6 NODOS



DEFINICION ISOPARAMETRICA DEL ELEMENTO

$$\begin{bmatrix} 1 \\ x \\ y \\ u_x \\ u_y \end{bmatrix} = \begin{bmatrix} 1 & 1 & 1 & 1 & 1 & 1 \\ x_1 & x_2 & x_3 & x_4 & x_5 & x_6 \\ y_1 & y_2 & y_3 & y_4 & y_5 & y_6 \\ u_{x1} & u_{x2} & u_{x3} & u_{x4} & u_{x5} & u_{x6} \\ u_{y1} & u_{y2} & u_{y3} & u_{y4} & u_{y5} & u_{y6} \end{bmatrix} \mathbf{N}$$

$$\mathbf{N} = \begin{bmatrix} N_1^{(e)} \\ N_2^{(e)} \\ N_3^{(e)} \\ N_4^{(e)} \\ N_5^{(e)} \\ N_6^{(e)} \end{bmatrix} = \begin{bmatrix} \zeta_1(2\zeta_1-1) \\ \zeta_2(2\zeta_2-1) \\ \zeta_3(2\zeta_3-1) \\ 4\zeta_1\zeta_2 \\ 4\zeta_2\zeta_3 \\ 4\zeta_3\zeta_1 \end{bmatrix}, \quad \frac{\partial \mathbf{N}}{\partial \zeta_1} = \begin{bmatrix} 4\zeta_1 - 1 \\ 0 \\ 0 \\ 4\zeta_2 \\ 0 \\ 4\zeta_3 \end{bmatrix}, \quad \frac{\partial \mathbf{N}}{\partial \zeta_2} = \begin{bmatrix} 0 \\ 4\zeta_2 - 1 \\ 0 \\ 4\zeta_1 \\ 4\zeta_3 \\ 0 \end{bmatrix}, \quad \frac{\partial \mathbf{N}}{\partial \zeta_3} = \begin{bmatrix} 0 \\ 0 \\ 4\zeta_3 - 1 \\ 0 \\ 4\zeta_2 \\ 4\zeta_1 \end{bmatrix}$$

CALCULO DE UNA DERIVADA PARCIAL GENERICA

The bulk of the shape function subroutine is concerned with the computation of the partial derivatives of the shape functions (24.12) with respect to x and y at any point in the element. For this purpose consider a generic scalar function: $w(\zeta_1, \zeta_2, \zeta_3)$, which is quadratically interpolated over the triangle by

$$w = [\, w_1 \quad w_2 \quad w_3 \quad w_4 \quad w_5 \quad w_6 \,] \begin{bmatrix} \zeta_1(2\zeta_1-1) \\ \zeta_2(2\zeta_2-1) \\ \zeta_3(2\zeta_3-1) \\ 4\zeta_1\zeta_2 \\ 4\zeta_2\zeta_3 \\ 4\zeta_3\zeta_1 \end{bmatrix}$$

Symbol w may represent 1, x, y, u_x or u_y, which are interpolated in the iso-P representation (24.11), or other element-varying quantities such as thickness, temperature, etc.

$$\frac{\partial w}{\partial x} = \sum w_i \frac{\partial N_i}{\partial x} = \sum w_i \left(\frac{\partial N_i}{\partial \zeta_1}\frac{\partial \zeta_1}{\partial x} + \frac{\partial N_i}{\partial \zeta_2}\frac{\partial \zeta_2}{\partial x} + \frac{\partial N_i}{\partial \zeta_3}\frac{\partial \zeta_3}{\partial x} \right)$$

$$\frac{\partial w}{\partial y} = \sum w_i \frac{\partial N_i}{\partial y} = \sum w_i \left(\frac{\partial N_i}{\partial \zeta_1}\frac{\partial \zeta_1}{\partial y} + \frac{\partial N_i}{\partial \zeta_2}\frac{\partial \zeta_2}{\partial y} + \frac{\partial N_i}{\partial \zeta_3}\frac{\partial \zeta_3}{\partial y} \right)$$

$$\begin{bmatrix} \frac{\partial w}{\partial x} \\ \frac{\partial w}{\partial y} \end{bmatrix} = \begin{bmatrix} \frac{\partial \zeta_1}{\partial x} & \frac{\partial \zeta_2}{\partial x} & \frac{\partial \zeta_3}{\partial x} \\ \frac{\partial \zeta_1}{\partial y} & \frac{\partial \zeta_2}{\partial y} & \frac{\partial \zeta_3}{\partial y} \end{bmatrix} \begin{bmatrix} \sum w_i \frac{\partial N_i}{\partial \zeta_1} \\ \sum w_i \frac{\partial N_i}{\partial \zeta_2} \\ \sum w_i \frac{\partial N_i}{\partial \zeta_3} \end{bmatrix}$$

$$\left[\sum w_i \frac{\partial N_i}{\partial \zeta_1} \quad \sum w_i \frac{\partial N_i}{\partial \zeta_2} \quad \sum w_i \frac{\partial N_i}{\partial \zeta_3}\right] \begin{bmatrix} \frac{\partial \zeta_1}{\partial x} & \frac{\partial \zeta_1}{\partial y} \\ \frac{\partial \zeta_2}{\partial x} & \frac{\partial \zeta_2}{\partial y} \\ \frac{\partial \zeta_3}{\partial x} & \frac{\partial \zeta_3}{\partial y} \end{bmatrix} = \left[\frac{\partial w}{\partial x} \quad \frac{\partial w}{\partial y}\right]$$

INTRODUCIENDO FILAS CORRESPONDIENTES A GEOMETRIA

$$\begin{bmatrix} \sum \frac{\partial N_i}{\partial \zeta_1} & \sum \frac{\partial N_i}{\partial \zeta_2} & \sum \frac{\partial N_i}{\partial \zeta_3} \\ \sum x_i \frac{\partial N_i}{\partial \zeta_1} & \sum x_i \frac{\partial N_i}{\partial \zeta_2} & \sum x_i \frac{\partial N_i}{\partial \zeta_3} \\ \sum y_i \frac{\partial N_i}{\partial \zeta_1} & \sum y_i \frac{\partial N_i}{\partial \zeta_2} & \sum y_i \frac{\partial N_i}{\partial \zeta_3} \end{bmatrix} \begin{bmatrix} \frac{\partial \zeta_1}{\partial x} & \frac{\partial \zeta_1}{\partial y} \\ \frac{\partial \zeta_2}{\partial x} & \frac{\partial \zeta_2}{\partial y} \\ \frac{\partial \zeta_3}{\partial x} & \frac{\partial \zeta_3}{\partial y} \end{bmatrix} = \begin{bmatrix} \frac{\partial 1}{\partial x} & \frac{\partial 1}{\partial y} \\ \frac{\partial x}{\partial x} & \frac{\partial x}{\partial y} \\ \frac{\partial y}{\partial x} & \frac{\partial y}{\partial y} \end{bmatrix}$$

But $\partial x/\partial x = \partial y/\partial y = 1$ and $\partial 1/\partial x = \partial 1/\partial y = \partial x/\partial y = \partial y/\partial x = 0$ because x and y are independent coordinates. It is shown in Remark 24.2 below that, if $\sum N_i = 1$, the entries of the first row of the coefficient matrix are equal to a constant C. These can be scaled to unity because the first row of the right-hand side is null. Consequently we arrive at a system of linear equations of order 3 with two right-hand sides:

SISTEMA DE ECUACIONES LINEALES A RESOLVER

$$\begin{bmatrix} 1 & 1 & 1 \\ \sum x_i \frac{\partial N_i}{\partial \zeta_1} & \sum x_i \frac{\partial N_i}{\partial \zeta_2} & \sum x_i \frac{\partial N_i}{\partial \zeta_3} \\ \sum y_i \frac{\partial N_i}{\partial \zeta_1} & \sum y_i \frac{\partial N_i}{\partial \zeta_2} & \sum y_i \frac{\partial N_i}{\partial \zeta_3} \end{bmatrix} \begin{bmatrix} \frac{\partial \zeta_1}{\partial x} & \frac{\partial \zeta_1}{\partial y} \\ \frac{\partial \zeta_2}{\partial x} & \frac{\partial \zeta_2}{\partial y} \\ \frac{\partial \zeta_3}{\partial x} & \frac{\partial \zeta_3}{\partial y} \end{bmatrix} = \begin{bmatrix} 0 & 0 \\ 1 & 0 \\ 0 & 1 \end{bmatrix}$$

PARA ELEMENTO CUADRATICO DE 6 NODOS

$$\begin{bmatrix} 1 & 1 & 1 \\ x_1(4\zeta_1 - 1) + 4x_4\zeta_2 + 4x_6\zeta_3 & x_2(4\zeta_2 - 1) + 4x_5\zeta_3 + 4x_4\zeta_1 & x_3(4\zeta_3 - 1) + 4x_6\zeta_1 + 4x_5\zeta_2 \\ y_1(4\zeta_1 - 1) + 4y_4\zeta_2 + 4y_6\zeta_3 & y_2(4\zeta_2 - 1) + 4y_5\zeta_3 + 4y_4\zeta_1 & y_3(4\zeta_3 - 1) + 4y_6\zeta_1 + 4y_5\zeta_2 \end{bmatrix} \begin{bmatrix} \frac{\partial \zeta_1}{\partial x} & \frac{\partial \zeta_1}{\partial y} \\ \frac{\partial \zeta_2}{\partial x} & \frac{\partial \zeta_2}{\partial y} \\ \frac{\partial \zeta_3}{\partial x} & \frac{\partial \zeta_3}{\partial y} \end{bmatrix} = \begin{bmatrix} 0 & 0 \\ 1 & 0 \\ 0 & 1 \end{bmatrix}$$

$$\mathbf{JP} = \mathbf{R}$$

By analogy with the quadrilateral isoparametric elements, the coefficient matrix of system (24.18) is called the *Jacobian matrix* and is denoted by **J**. Its determinant scaled by one half is equal to the Jacobian $J = \frac{1}{2}\det\mathbf{J}$ used in the expression of the area element introduced in §24.2.3. For compactness (24.18) is

$$\mathbf{J}\,\mathbf{P} = \begin{bmatrix} 1 & 1 & 1 \\ J_{x1} & J_{x2} & J_{x3} \\ J_{y1} & J_{y2} & J_{y3} \end{bmatrix} \begin{bmatrix} \frac{\partial\zeta_1}{\partial x} & \frac{\partial\zeta_1}{\partial y} \\ \frac{\partial\zeta_2}{\partial x} & \frac{\partial\zeta_2}{\partial y} \\ \frac{\partial\zeta_3}{\partial x} & \frac{\partial\zeta_3}{\partial y} \end{bmatrix} = \begin{bmatrix} 0 & 0 \\ 1 & 0 \\ 0 & 1 \end{bmatrix}$$

SOLUCION

$$\begin{bmatrix} \frac{\partial\zeta_1}{\partial x} & \frac{\partial\zeta_1}{\partial y} \\ \frac{\partial\zeta_2}{\partial x} & \frac{\partial\zeta_2}{\partial y} \\ \frac{\partial\zeta_3}{\partial x} & \frac{\partial\zeta_3}{\partial y} \end{bmatrix} = \frac{1}{2J}\begin{bmatrix} J_{y23} & J_{x32} \\ J_{y31} & J_{x13} \\ J_{y12} & J_{x21} \end{bmatrix} = \mathbf{P}$$

$J_{xji} = J_{xj} - J_{xi}$, $J_{yji} = J_{yj} - J_{yi}$ and $J = \frac{1}{2}\det\mathbf{J} = \frac{1}{2}(J_{x21}J_{y31} - J_{y12}J_{x13})$

SOLUCION DE UNA DERIVADA PARCIAL GENERICA

$$\frac{\partial w}{\partial x} = \sum w_i \frac{\partial N_i}{\partial x} = \sum \frac{w_i}{2J}\left(\frac{\partial N_i}{\partial\zeta_1}J_{y23} + \frac{\partial N_i}{\partial\zeta_2}J_{y31} + \frac{\partial N_i}{\partial\zeta_3}J_{y12}\right)$$

$$\frac{\partial w}{\partial y} = \sum w_i \frac{\partial N_i}{\partial y} = \sum \frac{w_i}{2J}\left(\frac{\partial N_i}{\partial\zeta_1}J_{x32} + \frac{\partial N_i}{\partial\zeta_2}J_{x13} + \frac{\partial N_i}{\partial\zeta_3}J_{x21}\right)$$

SOLUCION DERIVADAS PARCIALES FUNCIONES DE FORMA

$$\frac{\partial N_i}{\partial x} = \frac{1}{2J}\left(\frac{\partial N_i}{\partial\zeta_1}J_{y23} + \frac{\partial N_i}{\partial\zeta_2}J_{y31} + \frac{\partial N_i}{\partial\zeta_3}J_{y12}\right)$$

$$\frac{\partial N_i}{\partial y} = \frac{1}{2J}\left(\frac{\partial N_i}{\partial\zeta_1}J_{x32} + \frac{\partial N_i}{\partial\zeta_2}J_{x13} + \frac{\partial N_i}{\partial\zeta_3}J_{x21}\right)$$

$$\begin{bmatrix} \frac{\partial N_i}{\partial x} \\ \frac{\partial N_i}{\partial y} \end{bmatrix} = \mathbf{P}^T \begin{bmatrix} \frac{\partial N_i}{\partial\zeta_1} & \frac{\partial N_i}{\partial\zeta_2} & \frac{\partial N_i}{\partial\zeta_3} \end{bmatrix}^T$$

001 APLICACIÓN AL TRIANGULO CUADRATICO CURSO 2004-5

MODULO FUNCIONES DE FORMA

ELEMENTOS MATRIZ JACOBIANA

$$J_{x1} = x_1(4\zeta_1 - 1) + 4(x_4\zeta_2 + x_6\zeta_3),\ \ J_{x2} = x_2(4\zeta_2 - 1) + 4(x_5\zeta_3 + x_4\zeta_1),\ \ J_{x3} = x_3(4\zeta_3 - 1) + 4(x_6\zeta_1 + x_5\zeta_2)$$
$$J_{y1} = y_1(4\zeta_1 - 1) + 4(y_4\zeta_2 + y_6\zeta_3),\ \ J_{y2} = y_2(4\zeta_2 - 1) + 4(y_5\zeta_3 + y_4\zeta_1),\ \ J_{y3} = y_3(4\zeta_3 - 1) + 4(y_6\zeta_1 + y_5\zeta_2)$$

COORDENADAS “JERARQUICAS” QUE PERMITEN SIMPLIFICAR EXPRESIONES

$$\Delta x_4 = x_4 - \tfrac{1}{2}(x_1 + x_2), \quad \Delta x_5 = x_5 - \tfrac{1}{2}(x_2 + x_3), \quad \Delta x_6 = x_6 - \tfrac{1}{2}(x_3 + x_1)$$
$$\Delta y_4 = y_4 - \tfrac{1}{2}(y_1 + y_2), \quad \Delta y_5 = y_5 - \tfrac{1}{2}(y_2 + y_3), \quad \Delta y_6 = y_6 - \tfrac{1}{2}(y_3 + y_1)$$

Geometrically these represent the deviations from the midpoint positions; thus for a superparametric element $\Delta x4 = \Delta x5 = \Delta x6 = \Delta y4 = \Delta y5 = \Delta y6 = 0$.

ELEMENTOS MATRIZ JACOBIANA (CONT.)

$$J_{x21} = x_{21} + 4(\Delta x_4(\zeta_1-\zeta_2) + (\Delta x_5-\Delta x_6)\zeta_3), \quad J_{x32} = x_{32} + 4(\Delta x_5(\zeta_2-\zeta_3) + (\Delta x_6-\Delta x_4)\zeta_1)$$
$$J_{x13} = x_{13} + 4(\Delta x_6(\zeta_3-\zeta_1) + (\Delta x_4-\Delta x_5)\zeta_2), \quad J_{y12} = y_{12} + 4(\Delta y_4(\zeta_2-\zeta_1) + (\Delta y_6-\Delta y_5)\zeta_3)$$
$$J_{y23} = y_{23} + 4(\Delta y_5(\zeta_3-\zeta_2) + (\Delta y_4-\Delta y_6)\zeta_1), \quad J_{y31} = y_{31} + 4(\Delta y_6(\zeta_1-\zeta_3) + (\Delta y_5-\Delta y_4)\zeta_2)$$

JACOBIANO

$$J = \tfrac{1}{2}\det \mathbf{J} = \tfrac{1}{2}(J_{x21}J_{y31} - J_{y12}J_{x13})$$

SOLUCION DERIVADAS FUNCIONES DE FORMA

$$\mathbf{P}^T = \frac{1}{2J}\begin{bmatrix} y_{23} + 4(\Delta y_5(\zeta_3-\zeta_2) + (\Delta y_4-\Delta y_6)\zeta_1) & x_{32} + 4(\Delta x_5(\zeta_2-\zeta_3) + (\Delta x_6-\Delta x_4)\zeta_1) \\ y_{31} + 4(\Delta y_6(\zeta_1-\zeta_3) + (\Delta y_5-\Delta y_4)\zeta_2) & x_{13} + 4(\Delta x_6(\zeta_3-\zeta_1) + (\Delta x_4-\Delta x_5)\zeta_2) \\ y_{12} + 4(\Delta y_4(\zeta_2-\zeta_1) + (\Delta y_6-\Delta y_5)\zeta_3) & x_{21} + 4(\Delta x_4(\zeta_1-\zeta_2) + (\Delta x_5-\Delta x_6)\zeta_3) \end{bmatrix}$$

DERIVADAS CARTESIANAS FUNCIONES DE FORMA

$$\frac{\partial \mathbf{N}}{\partial x} = \frac{1}{2J}\begin{bmatrix} (4\zeta_1 - 1)J_{y23} \\ (4\zeta_2 - 1)J_{y31} \\ (4\zeta_3 - 1)J_{y12} \\ 4(\zeta_2 J_{y23} + \zeta_1 J_{y31}) \\ 4(\zeta_3 J_{y31} + \zeta_2 J_{y12}) \\ 4(\zeta_1 J_{y12} + \zeta_3 J_{y23}) \end{bmatrix}, \quad \frac{\partial \mathbf{N}}{\partial y} = \frac{1}{2J}\begin{bmatrix} (4\zeta_1 - 1)J_{x32} \\ (4\zeta_2 - 1)J_{x13} \\ (4\zeta_3 - 1)J_{x21} \\ 4(\zeta_2 J_{x32} + \zeta_1 J_{x13}) \\ 4(\zeta_3 J_{x13} + \zeta_2 J_{x21}) \\ 4(\zeta_1 J_{x21} + \zeta_3 J_{x32}) \end{bmatrix}$$

001 TRIANGULO 6 – MODULO FUNCIONES FORMA CURSO 2004-5

```
Trig6IsoPShapeFunDer[ncoor_,tcoor_]:= Module[
  {ζ1,ζ2,ζ3,x1,x2,x3,x4,x5,x6,y1,y2,y3,y4,y5,y6,
  dx4,dx5,dx6,dy4,dy5,dy6,Jx21,Jx32,Jx13,Jy12,Jy23,Jy31,
  Nf,dNx,dNy,Jdet}, {ζ1,ζ2,ζ3}=tcoor;
  {{x1,y1},{x2,y2},{x3,y3},{x4,y4},{x5,y5},{x6,y6}}=ncoor;
  dx4=x4-(x1+x2)/2; dx5=x5-(x2+x3)/2; dx6=x6-(x3+x1)/2;
  dy4=y4-(y1+y2)/2; dy5=y5-(y2+y3)/2; dy6=y6-(y3+y1)/2;
  Nf={ζ1*(2*ζ1-1),ζ2*(2*ζ2-1),ζ3*(2*ζ3-1),4*ζ1*ζ2,4*ζ2*ζ3,4*ζ3*ζ1};
  Jx21= x2-x1+4*(dx4*(ζ1-ζ2)+(dx5-dx6)*ζ3);
  Jx32= x3-x2+4*(dx5*(ζ2-ζ3)+(dx6-dx4)*ζ1);
  Jx13= x1-x3+4*(dx6*(ζ3-ζ1)+(dx4-dx5)*ζ2);
  Jy12= y1-y2+4*(dy4*(ζ2-ζ1)+(dy6-dy5)*ζ3);
  Jy23= y2-y3+4*(dy5*(ζ3-ζ2)+(dy4-dy6)*ζ1);
  Jy31= y3-y1+4*(dy6*(ζ1-ζ3)+(dy5-dy4)*ζ2);
  Jdet = Jx21*Jy31-Jy12*Jx13;
  dNx= {(4*ζ1-1)*Jy23,(4*ζ2-1)*Jy31,(4*ζ3-1)*Jy12,4*(ζ2*Jy23+ζ1*Jy31),
        4*(ζ3*Jy31+ζ2*Jy12),4*(ζ1*Jy12+ζ3*Jy23)}/Jdet;
  dNy= {(4*ζ1-1)*Jx32,(4*ζ2-1)*Jx13,(4*ζ3-1)*Jx21,4*(ζ2*Jx32+ζ1*Jx13),
        4*(ζ3*Jx13+ζ2*Jx21),4*(ζ1*Jx21+ζ3*Jx32)}/Jdet;
  Return[Simplify[{Nf,dNx,dNy,Jdet}]]
];
```

This module receives two arguments: `ncoor` and `tcoor`. The first one is the list of $\{x_i, y_i\}$ coordinates of the six nodes: `{{x1,y1},{x2,y2}, ... {x6,y6}}`. The second is the list of three triangular coordinates $\{\zeta_1, \zeta_2, \zeta_3\}$ of the location at which the shape functions and their Cartesian derivatives are to be computed.

The module returns `{Nf,dNx,dNy,Jdet}` as module value. Here `Nf` collects the shape function values, `dNx` the shape function x derivative values, `dNy` the shape function y derivative values, and `Jdet` is the determinant of matrix $\mathbf{J}$, equal to $2J$ in the notation used here.

MATRIZ DE RIGIDEZ

$$\mathbf{K}^{(e)} = \int_{\Omega^{(e)}} h\mathbf{B}^T\mathbf{E}\mathbf{B}\, d\Omega^{(e)} \approx \sum_{i=1}^{p} w_i \mathbf{F}(\zeta_{1i}, \zeta_{2i}, \zeta_{3i}), \qquad \text{where} \quad \mathbf{F}(\zeta_1, \zeta_2, \zeta_3) = h\mathbf{B}^T\mathbf{E}\mathbf{B}\, J.$$

Here p denotes the number of sample points of the Gauss rule being used, w_i is the integration weight for the i^{th} sample point, $\zeta_{1i}, \zeta_{2i}, \zeta_{3i}$ are the sample point triangular coordinates and $J = \frac{1}{2}\det\mathbf{J}$. The last four numbers are returned by `TrigGaussRuleInfo` as explained in the previous section.

```
Trig6IsoPMembraneStiffness[ncoor_,mprop_,fprop_,opt_]:=
 Module[{i,k,l,p=3,num=False,Emat,th={fprop},h,tcoor,w,c,
  Nf,dNx,dNy,Jdet,Be,Ke=Table[0,{12},{12}]},
  Emat=mprop[[1]]; If [Length[fprop]>0, th=fprop[[1]]];
  If [Length[opt]>0,   num=opt[[1]]];
  If [Length[opt]>1,   p=  opt[[2]]];
  If [p!=-3&&p!=1&&p!=3&&p!=7, Print["Illegal p"];Return[Null]];
  For [k=1, k<=Abs[p], k++,
      {tcoor,w}=          TrigGaussRuleInfo[{p,num},k];
      {Nf,dNx,dNy,Jdet}= Trig6IsoPShapeFunDer[ncoor,tcoor];
      If [Length[th]==0, h=th, h=th.Nf]; c=w*Jdet*h/2;
      Be={Flatten[Table[{dNx[[i]],       0},{i,6}]],
          Flatten[Table[{0,       dNy[[i]]},{i,6}]],
          Flatten[Table[{dNy[[i]],dNx[[i]]},{i,6}]]};
      Ke+=c*Transpose[Be].(Emat.Be);
      ]; Return[Ke]
   ];
```

Module `Trig6IsoPMembraneStiffness`, listed in Figure 24.6, implements the computation of the element stiffness matrix of a quadratic plane stress triangle. The arguments of the module are

`ncoor` Node coordinates arranged in two-dimensional list form: `{{x1,y1},{x2,y2},{x3,y3},{x4,y4},{x5,y5},{x6,y6}}`.

`mprop` Material properties supplied as the list `{Emat,rho,alpha}`. `Emat` is a two-dimensional list storing the 3 × 3 plane stress matrix of elastic moduli:

$$\mathbf{E} = \begin{bmatrix} E_{11} & E_{12} & E_{13} \\ E_{12} & E_{22} & E_{23} \\ E_{13} & E_{23} & E_{33} \end{bmatrix} \qquad (24.32)$$

The other two items in `mprop` are not used in this module so zeros may be inserted as placeholders.

`fprop` Fabrication properties. The plate thickness specified as a one-entry list: `{h}`, as a six-entry list: `{h1,h2,h3,h4,h5,h6}`, or as an empty list: `{ }`.

The one-entry form specifies uniform thickness `h`. The six-entry form is used to specify an element of variable thickness, in which case the entries are the six node thicknesses and h is interpolated quadratically. If an empty list appears the module assumes a uniform unit thickness.

`options` Processing options. This list may contain two items: `{numer,rule}` or one: `{numer}`.

`numer` is a flag with value `True` or `False`. If `True`, the computations are forced to go in floating point arithmetic. For symbolic or exact arithmetic work set `numer` to `False`.

`rule` specifies the triangle Gauss rule as described in §24.2.4. `rule` may be 1, 3, −3, 6 or 7. For the 6-node element the three point rules are sufficient to get the correct rank. If omitted `rule` = 3 is assumed.

001 MODELOS COMPROBACION MODULOS CURSO 2004-5

The stiffness module is tested on the two triangle geometries shown in Figure 24.7. Both elements have unit thickness and isotropic material. The left one has the corner nodes placed at (0, 0), (4, 2) and 6, 4) with side nodes 4,5,6 at the midpoints of the sides. The right one has the 3 corners forming an equilateral triangle: $-1/2$, 0, 1/2, 0, 0, $\sqrt{3}/2$ whereas the side nodes are placed at 0, $-1/(2\sqrt{3})$, 1/2, $1/\sqrt{3}$, $-1/2$, $1/\sqrt{3}$ so that the six nodes lie on a circle of radius $1/\sqrt{3}$. These geometries will be used to illustrate the effect of the numerical integration rule.





001 TRIANGULO LADOS RECTOS CURSO 2004-5

$E = 288$, $\nu=1/3$ and $h = 1$

```
ClearAll[Em,nu,h];    h=1; Em=288; nu=1/3;
ncoor={{0,0},{6,2},{4,4},{3,1},{5,3},{2,2}};
Emat=Em/(1-nu^2)*{{1,nu,0},{nu,1,0},{0,0,(1-nu)/2}};
Ke=Trig6IsoPMembraneStiffness[ncoor,{Emat,0,0},{h},{False,-3}];
Ke=Simplify[Ke];  Print[Chop[Ke]//MatrixForm];
Print["eigs of Ke=",Chop[Eigenvalues[N[Ke]]]];
```

The returned stiffness matrix for integration `rule=3`, `rule=-3`, and `rule=7` is the same since those are exact for this element if the side nodes are at the midpoints, which is the case here. That stiffness is

MATRIZ RIGIDEZ

$$\begin{bmatrix} 54 & 27 & 18 & 0 & 0 & 9 & -72 & 0 & 0 & 0 & 0 & -36 \\ 27 & 54 & 0 & -18 & 9 & 36 & 0 & 72 & 0 & 0 & -36 & -144 \\ 18 & 0 & 216 & -108 & 54 & -36 & -72 & 0 & -216 & 144 & 0 & 0 \\ 0 & -18 & -108 & 216 & -36 & 90 & 0 & 72 & 144 & -360 & 0 & 0 \\ 0 & 9 & 54 & -36 & 162 & -81 & 0 & 0 & -216 & 144 & 0 & -36 \\ 9 & 36 & -36 & 90 & -81 & 378 & 0 & 0 & 144 & -360 & -36 & -144 \\ -72 & 0 & -72 & 0 & 0 & 0 & 576 & -216 & 0 & -72 & -432 & 288 \\ 0 & 72 & 0 & 72 & 0 & 0 & -216 & 864 & -72 & -288 & 288 & -720 \\ 0 & 0 & -216 & 144 & -216 & 144 & 0 & -72 & 576 & -216 & -144 & 0 \\ 0 & 0 & 144 & -360 & 144 & -360 & -72 & -288 & -216 & 864 & 0 & 144 \\ 0 & -36 & 0 & 0 & 0 & -36 & -432 & 288 & -144 & 0 & 576 & -216 \\ -36 & -144 & 0 & 0 & -36 & -144 & 288 & -720 & 0 & 144 & -216 & 864 \end{bmatrix}$$

VALORES PROPIOS

$$[\,1971.66 \;\; 1416.75 \;\; 694.82 \;\; 545.72 \;\; 367.7 \;\; 175.23 \;\; 157.68 \;\; 57.54 \;\; 12.899 \;\; 0 \;\; 0 \;\; 0\,]$$

The 3 zero eigenvalues pertain to the three independent rigid-body modes. The 9 other ones are positive. Consequently the computed $\mathbf{K}^{(e)}$ has the correct rank of 9.

$$E = 504, \nu = 0, h = 1$$

For the "circle" geometry the results for $E = 504$, $\nu = 0$, $h = 1$ and the rank-sufficient rules 3,-3,6,7 are obtained through the following script:

```
ClearAll[Em,nu,h]; Em=7*72; nu=0; h=1;
{x1,y1}={-1,0}/2; {x2,y2}={1,0}/2; {x3,y3}={0,Sqrt[3]}/2;
{x4,y4}={0,-1/Sqrt[3]}/2; {x5,y5}={1/2,1/Sqrt[3]}; {x6,y6}={-1/2,1/Sqrt[3]};
ncoor= {{x1,y1},{x2,y2},{x3,y3},{x4,y4},{x5,y5},{x6,y6}};
Emat=Em/(1-nu^2)*{{1,nu,0},{nu,1,0},{0,0,(1-nu)/2}};
For [i=2,i<=5,i++, p={1,-3,3,6,7}[[i]];
    Ke=Trig6IsoPMembraneStiffness[ncoor,{Emat,0,0},{h},{True,p}];
    Ke=Chop[Simplify[Ke]];
    Print["Ke=",SetPrecision[Ke,4]//MatrixForm];
    Print["Eigenvalues of Ke=",Chop[Eigenvalues[N[Ke]],.0000001]]
];
```

This is straightforward except perhaps for the SetPrecision[Ke,4] statement in the Print. This specifies that the stiffness matrix entries be printed to only 4 places to save space (the default is otherwise 6 places).

MATRIZ DE RIGIDEZ

For rule=3:

344.7	75.00	−91.80	21.00	−86.60	−24.00	−124.7	−72.00	−20.78	−36.00	−20.78	36.00
75.00	258.1	−21.00	−84.87	18.00	−90.07	96.00	0	−36.00	20.78	−132.0	−103.9
−91.80	−21.00	344.7	−75.00	−86.60	24.00	−124.7	72.00	−20.78	−36.00	−20.78	36.00
21.00	−84.87	−75.00	258.1	−18.00	−90.07	−96.00	0	132.0	−103.9	36.00	20.78
−86.60	18.00	−86.60	−18.00	214.8	0	41.57	0	−41.57	144.0	−41.57	−144.0
−24.00	−90.07	24.00	−90.07	0	388.0	0	−41.57	−24.00	−83.14	24.00	−83.14
−124.7	96.00	−124.7	−96.00	41.57	0	374.1	0	−83.14	−72.00	−83.14	72.00
−72.00	0	72.00	0	0	−41.57	0	374.1	−72.00	−166.3	72.00	−166.3
−20.78	−36.00	−20.78	132.0	−41.57	−24.00	−83.14	−72.00	374.1	0	−207.8	0
−36.00	20.78	−36.00	−103.9	144.0	−83.14	−72.00	−166.3	0	374.1	0	−41.57
−20.78	−132.0	−20.78	36.00	−41.57	24.00	−83.14	72.00	−207.8	0	374.1	0
36.00	−103.9	36.00	20.78	−144.0	−83.14	72.00	−166.3	0	−41.57	0	374.1

For rule=-3:

566.4	139.0	129.9	21.00	79.67	8.000	−364.9	−104.0	−205.5	−36.00	−205.5	−28.00
139.0	405.9	−21.00	62.93	50.00	113.2	64.00	−129.3	−36.00	−164.0	−196.0	−288.7
129.9	−21.00	566.4	−139.0	79.67	−8.000	−364.9	104.0	−205.5	28.00	−205.5	36.00
21.00	62.93	−139.0	405.9	−50.00	113.2	−64.00	−129.3	196.0	−288.7	36.00	−164.0
79.67	50.00	79.67	−50.00	325.6	0	−143.2	0	−170.9	176.0	−170.9	−176.0
8.000	113.2	−8.000	113.2	0	646.6	0	−226.3	8.000	−323.3	−8.000	−323.3
−364.9	64.00	−364.9	−64.00	−143.2	0	632.8	0	120.1	−104.0	120.1	104.0
−104.0	−129.3	104.0	−129.3	0	−226.3	0	485.0	−104.0	0	104.0	0
−205.5	−36.00	−205.5	196.0	−170.9	8.000	120.1	−104.0	521.9	−64.00	−60.04	0
−36.00	−164.0	28.00	−288.7	176.0	−323.3	−104.0	0	−64.00	595.8	0	180.1
−205.5	−196.0	−205.5	36.00	−170.9	−8.000	120.1	104.0	−60.04	0	521.9	64.00
−28.00	−288.7	36.00	−164.0	−176.0	−323.3	104.0	0	0	180.1	64.00	595.8

For rule=7

661.9	158.5	141.7	21.00	92.53	7.407	−432.1	−117.2	−190.2	−29.10	−273.8	−40.61
158.5	478.8	−21.00	76.13	49.41	125.3	50.79	−182.7	−29.10	−156.6	−208.6	−341.0
141.7	−21.00	661.9	−158.5	92.53	−7.407	−432.1	117.2	−273.8	40.61	−190.2	29.10
21.00	76.13	−158.5	478.8	−49.41	125.3	−50.79	−182.7	208.6	−341.0	29.10	−156.6
92.53	49.41	92.53	−49.41	387.3	0	−139.8	0	−216.3	175.4	−216.3	−175.4
7.407	125.3	−7.407	125.3	0	753.4	0	−207.0	7.407	−398.5	−7.407	−398.5
−432.1	50.79	−432.1	−50.79	−139.8	0	723.6	0	140.2	−117.2	140.2	117.2
−117.2	−182.7	117.2	−182.7	0	−207.0	0	562.6	−117.2	4.884	117.2	4.884
−190.2	−29.10	−273.8	208.6	−216.3	7.407	140.2	−117.2	602.8	−69.71	−62.78	0
−29.10	−156.6	40.61	−341.0	175.4	−398.5	−117.2	4.884	−69.71	683.3	0	207.9
−273.8	−208.6	−190.2	29.10	−216.3	−7.407	140.2	117.2	−62.78	0	602.8	69.71
−40.61	−341.0	29.10	−156.6	−175.4	−398.5	117.2	4.884	0	207.9	69.71	683.3

001 TRIANGULO LADOS CURVOS **CURSO 2004-5**

VALORES PROPIOS

Rule	Eigenvalues of $\mathbf{K}^{(e)}$											
3	702.83	665.11	553.472	553.472	481.89	429.721	429.721	118.391	118.391	0	0	0
−3	1489.80	1489.80	702.833	665.108	523.866	523.866	481.890	196.429	196.429	0	0	0
6	1775.53	1775.53	896.833	768.948	533.970	533.970	495.570	321.181	321.181	0	0	0
7	1727.11	1727.11	880.958	760.719	532.750	532.750	494.987	312.123	312.123	0	0	0

Since the metric of this element is very distorted near its boundary, the stiffness matrix entries and eigenvalues change substantially as the integration formulas are advanced from 3 to 6 and 7 points. However as can be seen the matrix remains rank-sufficient.

001 EJERCICIO 1 **CURSO 2004-5**

EXERCISE 24.1

[C:20] Write an element stiffness module and a shape function module for the 4-node "transition" iso-P triangular element with only one side node: 4, which is located between 1 and 2. See Figure E24.1. The shape functions are $N_1 = \zeta_1 - 2\zeta_1\zeta_2$, $N_2 = \zeta_2 - 2\zeta_1\zeta_2$, $N_3 = \zeta_3$ and $N_4 = 4\zeta_1\zeta_2$. Use 3 interior point integration `rule=3`. Test the element for the geometry of the triangle depicted on the left of Figure 24.7, removing nodes 5 and 6, and with $E = 2880$, $\nu = 1/3$ and $h = 1$. Report results for the stiffness matrix $\mathbf{K}^{(e)}$ and its 8 eigenvalues. Note: Notebook `Trig6Stiffness.nb` for the 6-node triangle, posted on the index of Chapter 24, may be used as "template" in support of this Exercise.

Partial results: $K_{11} = 1980$, $K_{18} = 1440$.



Figure E24.1. The 4-node transition triangle for Exercise 24.1.